

My new test document

This is a new document, to evaluate the capabilities of Maple. This will allow me to determine, for future reference, which or Maple and its competitor Mathcad is the most efficient documentation tool. To do so, I will try different features, and compare them with the competitor product.

Basic math

Math editing methods

There are 2 main editing methods :

- 1D-math ("traditional" Maple math input)
- 2D-math ("natural" math)

Example of 1D-Math:

$$\left[\begin{array}{l} > \mathbf{a:=sqrt(2);} \\ & \qquad \qquad \qquad a := \sqrt{2} \end{array} \right. \quad (1.1.1)$$

Example of 2D-Math:

$$\left[\begin{array}{l} > a := \sqrt{2} \\ & \qquad \qquad \qquad a := \sqrt{2} \end{array} \right. \quad (1.1.2)$$

The traditional 1D-math approach probably has the advantage of compatibility with older versions of the software. It is also easy to type, as the final result is similar to what is written. However, complex expressions will be harder to read. 2D-math is much closer to natural math as learnt at school, and therefore easier to understand. On the other hand, 2D-math needs either to be accessed by clics in the menu, or to be typed as 1D-math then converted (right click on equation -> 2D-math -> convert to -> 2D math), which makes it slightly longer to implement. Note that the 2D-math doesn't need columns/semi-columns at the end of the expression.

Mixing math and text

There are 2 types of documents in Maple:

- "documents"
- "worksheets"

A "worksheet" document consists of a succession of prompts in which the user types a succession of prompts in which the user will type formulas and calculations, as follows:

$$\left[\begin{array}{l} > \mathbf{a:=sqrt(2);} \\ & \qquad \qquad \qquad a := \sqrt{2} \end{array} \right. \quad (1.2.1)$$

```

> b:=1:
> a+b;

```

$$\sqrt{2} + 1 \quad (1.2.2)$$

A "document" is as the present sheet : it allows text descriptions, and mixing math and text. It can include prompts, but also have the formulas mixed with the text, as follows: $a := \sqrt{2} = \sqrt{2}$ and then $b := 1$: and finally $a + b$

$$\sqrt{2} + 1 \quad (1.2.3)$$

In both cases, it is possible to decide not to display the result (using ":" at the end of the formula). In case of not-in-prompt math, it is possible to chose to display the result of the formula either in line (case "a"), or on another line (case "a+b"). If the result is displayed in line, the result is not "numbered", and cannot be re-used as such in future calculations. When it is on a different line, it can:

```

> 2·(1.2.3)

```

$$2\sqrt{2} + 2 \quad (1.2.4)$$

It is also possible to have an operation and its result separated by some text. For instance, the result of $2 \cdot a + b$ is $2\sqrt{2} + 1$.

Although the default look of a "worksheet" document is a succession of prompts, it is possible to "convert" all or part of it into a "document" looking sheet, and vice versa : there is no fundamental distinction between both, the only difference lies in the default settings of the document in terms of presentation.

Units

Maple is capable of managing units, which is absolutely essential to engineering documents. It includes a large set of SI units, as well as other less "scientific" ones. Units can be added to any variable. Units don't simplify automatically, but can simplify upon request, unless the unit package is included in the sheet

```

> R := 10 [[Ω]]

```

$$R := 10 \text{ [[Ω]]} \quad (1.3.1)$$

```

> Ia := 10 [[A]]

```

$$I_a := 10 \text{ [[A]]} \quad (1.3.2)$$

```

> Va := R·Ia

```

$$V_a := 100 \text{ [[V]]} \quad (1.3.3)$$

```

> simplify((1.3.3))

```

$$100 \text{ [[V]]} \quad (1.3.4)$$

Loading [Units:-Standard](#)

```

> Ib := 5 [[A]]

```

$$I_b := 5 \text{ [[A]]} \quad (1.3.5)$$

$$\begin{array}{l} \text{[> } V_b := R \cdot I_b \\ \qquad \qquad \qquad \qquad \qquad \qquad V_b := 50 \text{ [V]} \end{array} \quad (1.3.6)$$

Units are also available in 1D-math mode:

$$\begin{array}{l} \text{[> } I_c := 3 \cdot \text{Unit(A)}; \\ \qquad \qquad \qquad \qquad \qquad \qquad I_c := 3 \text{ [A]} \end{array} \quad (1.3.7)$$

It is also possible to create personalised units, as needed.

Advanced math

Functions & calculus

Defining functions is done as follows:

$$\begin{array}{l} \text{[> } f := x \rightarrow 3 \cdot x + 2 \\ \qquad \qquad \qquad \qquad \qquad \qquad f := x \rightarrow 3x + 2 \end{array} \quad (2.1.1)$$

$$\begin{array}{l} \text{[> } g := (x, y) \rightarrow x^2 + x \cdot y + y^2 \\ \qquad \qquad \qquad \qquad \qquad \qquad g := (x, y) \rightarrow x^2 + xy + y^2 \end{array} \quad (2.1.2)$$

$$\begin{array}{l} \text{[> } V_x := I_x \rightarrow I_x \cdot R \\ \qquad \qquad \qquad \qquad \qquad \qquad V_x := I_x \rightarrow I_x R \end{array} \quad (2.1.3)$$

Functions can be evaluated in different points:

$$\begin{array}{l} \text{[> } f(2) \\ \qquad \qquad \qquad \qquad \qquad \qquad 8 \end{array} \quad (2.1.4)$$

$$\begin{array}{l} \text{[> } g(1, 1) \\ \qquad \qquad \qquad \qquad \qquad \qquad 3 \end{array} \quad (2.1.5)$$

$$\begin{array}{l} \text{[> } V_x(3 \text{ [A]}) \\ \qquad \qquad \qquad \qquad \qquad \qquad 30 \text{ [V]} \end{array} \quad (2.1.6)$$

Operations such as derivation (and partial derivation) and integration, can be done on these functions:

$$\begin{array}{l} \text{[> } f'(x) \\ \qquad \qquad \qquad \qquad \qquad \qquad 3 \end{array} \quad (2.1.7)$$

$$\begin{array}{l} \text{[> } \frac{\partial}{\partial y} g(x, y) \\ \qquad \qquad \qquad \qquad \qquad \qquad x + 2y \end{array} \quad (2.1.8)$$

$$\begin{aligned} > \int_0^8 f(x) \, dx &= 112 & (2.1.9) \end{aligned}$$

$$\begin{aligned} > \int_3^t f(x) \, dx &= \frac{3}{2} t^2 - \frac{39}{2} + 2 t & (2.1.10) \end{aligned}$$

Vectors and matrix

Manipulating matrix requires the linear algebra package:

Loading [LinearAlgebra](#)

Vectors and matrix can be defined easily through the matrix wizzard, element-by-element, or as a whole with pre-defined functions:

$$\begin{aligned} > AA := \begin{bmatrix} 1 & 2 \\ 3 & 4 \\ 5 & 6 \end{bmatrix} & & AA := \begin{bmatrix} 1 & 2 \\ 3 & 4 \\ 5 & 6 \end{bmatrix} & (2.2.1) \end{aligned}$$

$$\begin{aligned} > BB := Matrix([[1, 2, 3], [4, 5, 6]]) & & BB := \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix} & (2.2.2) \end{aligned}$$

$$\begin{aligned} > ID := Matrix(2, shape = identity) & & ID := \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix} & (2.2.3) \end{aligned}$$

Operations can be done on matrix:

$$\begin{aligned} > CC := BB.AA & & CC := \begin{bmatrix} 22 & 28 \\ 49 & 64 \end{bmatrix} & (2.2.4) \end{aligned}$$

$$\begin{aligned} > Transpose(AA) & & \begin{bmatrix} 1 & 3 & 5 \\ 2 & 4 & 6 \end{bmatrix} & (2.2.5) \end{aligned}$$

Matrix are different from vectors, but can be manipulated in a similar way:

```
> DD := Matrix([[1], [2], [3]])
```

$$DD := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix} \quad (2.2.6)$$

```
> whattype(DD)
```

Matrix (2.2.7)

```
> EE := Vector([1, 2, 3])
```

$$EE := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix} \quad (2.2.8)$$

```
> whattype(EE)
```

Vector_{column} (2.2.9)

Series and Recursion

Using Matrix, it is very easy to define recursively a system.

```
> N := 100
```

$$N := 100 \quad (2.3.1)$$

```
> Rec := Matrix(N, 2, fill = 0)
```

$$Rec := \begin{bmatrix} 100 \times 2 \text{ Matrix} \\ \text{Data Type: anything} \\ \text{Storage: rectangular} \\ \text{Order: Fortran_order} \end{bmatrix} \quad (2.3.2)$$

```
> Rec[1, 1] := 1;
   Rec[1, 2] := 2;
```

$$\begin{aligned} Rec_{1,1} &:= 1 \\ Rec_{1,2} &:= 2 \end{aligned} \quad (2.3.3)$$

```
> for i from 2 to N do
   Rec[i, 1] := -1 * Rec[i - 1, 2];
   Rec[i, 2] := -1 * Rec[i - 1, 1];
end do;
```

$$\begin{aligned} &-2 \\ &-1 \end{aligned} \quad (2.3.4)$$

```
> Rec[N - 1, 1];
```

```
Rec[N - 1, 2];
```

```
1
2
```

(2.3.5)

Laplace transform

With the package `inttrans`, it is possible to manipulate transfer function, as well as to do laplace or fourier transform on time-domain function, as well as the inverse functions.

```
> with(inttrans)
[addtable, fourier, fouriercos, fouriersin, hankel, hilbert, invfourier, invhilbert,
 invlaplace, invmellin, laplace, mellin, savetable]
```

(2.4.1)

```
> TimeDomain := t -> sin(100 * pi * t) * e^(-t/0.03)
TimeDomain := t -> sin(100 * pi * t) * e^((-1) * t * Units:-Standard:-^(0.03))
```

(2.4.2)

```
> FrequencyDomain := laplace(TimeDomain(t), t, s)
FrequencyDomain := 3.141592654 * 10^18 / ((1.000000000 * 10^8 * s + 3.333333333 * 10^9)^2 + 9.869604404 * 10^20)
```

(2.4.3)

```
> FrequencyDomain(s)
3.141592654 * 10^18 / ((1.000000000 * 10^8 * s(s) + 3.333333333 * 10^9)^2 + 9.869604404 * 10^20)
```

(2.4.4)

Equation solving

Maple is capable of solving equations both numerically and symbolically:

```
> solve(x^2 = 5, x)
sqrt(5), -sqrt(5)
```

(2.5.1)

```
> solve(3 * x^2 + x + u = 0, x)
-1/6 + 1/6 * sqrt(1 - 12 * u), -1/6 - 1/6 * sqrt(1 - 12 * u)
```

(2.5.2)

It is possible to affect the solution(s) of the equation to a vector:

```
> XX := [solve(3 * x^2 + 2 * x - 5 = 0, x)]
XX := [1, -5/3]
```

(2.5.3)

```
> XX[2]
```

(2.5.4)

$$-\frac{5}{3} \quad (2.5.4)$$

It is also possible to directly assign the results to the variables that are used in the equation (here, A_x and A_y).

```

> unassign('A_x','A_y') :
  assign(solve({A_x + A_y = 12, 3·A_x - 5 A_y = 2}, {A_x, A_y}))
> A_x

```

$$\frac{31}{4} \quad (2.5.5)$$

```

> A_y

```

$$\frac{17}{4} \quad (2.5.6)$$

It is also possible to define the solution as a function of some parameters:

```

> SS := x → solve(3 x = x^2 + y + 12, y) :
> SS(x)

```

$$-x^2 + 3x - 12 \quad (2.5.7)$$

```

> SS(5)

```

$$-22 \quad (2.5.8)$$

▼ Differential equation solving

It is possible to solve differential equations with Maple.

```

> dsolve( { C_x · d/dt V_out(t) = (V_in - V_out(t))/R_x, V_out(0) = 0 } )

```

$$V_{out}(t) = V_{in} - e^{-\frac{t}{R_x C_x}} V_{in} \quad (2.6.1)$$

Differential equations can be solved literally, numerically, include units, etc.

▼ Code editing

Maple allows to write procedures in more code-like environments. Code regions can be collapsed.

```

1 MyProcedure:=proc(X,Y::positive,Z::expects(integer):=0)
2
3     local OUT;
4
5     if X>0 then
6         OUT:=X^2+Y+sqrt(Z);
7     else
8         OUT:=sqrt(X+Y+Z);
9     end if;
10
11     return OUT;
12
13 end proc:

```

In this example, we can see that the usual elements of programming (conditions, loops...) are available. It is also possible to indicate to the procedure the type of data expected as parameters, and default values. Default values allow to "skip" the use of a parameter if its value is the default one.

Usage where the last parameter is omitted (use of default value):

```
[> MyProcedure(2, 1);
```

5

(2.7.1)

Usage with all parameters

```
[> MyProcedure(2, 1, 3)
```

$5 + \sqrt{3}$

(2.7.2)

Usage with one parameter not of the expected type:

```
[> MyProcedure(2, -1, 3)
Error, invalid input: MyProcedure expects its 2nd argument,
Y, to be of type positive, but received -1
```

▼ Tolerance calculation

Maple has a built-in module to manage tolerances. Tolerances are compatible with the use of units.

Loading [Tolerances](#)

```
[> Rl := (100&+-5) [[Ω]]
```

(2.8.1)

$$R_1 := (100. \pm 5.) \llbracket \Omega \rrbracket \quad (2.8.1)$$

$$R_2 := (500. \pm 20.) \llbracket \Omega \rrbracket \quad (2.8.2)$$

$$R_{eq} := R_1 + R_2$$

$$R_{eq} := (600. \pm 25.) \llbracket \Omega \rrbracket \quad (2.8.3)$$

Plots & graphs


"Regular" plots

Loading [plots](#)


Using the plots package, it is possible to plot functions. It is necessary to use a command line to do so. Some parameters can be set through the GUI, but some of them will revert to default if the sheet is recalculated, if not included in the command line.

```
> f1 := x → x :
  f2 := x → 2 · x + 3 :
> plot( [ f1(x), f2(x) ], x = -5 .. 8, legend = ["function1", "function2"], title = "Plot example",
  linestyle = ["solid", "dash"], legendstyle = [location = right], gridlines = true)
```

Plot example



It is also possible to insert directly using the GUI, without typing its command-line. In which case, graph properties are "maintained" after recalculation.




It is also possible to plot scatters using vectors.

```
> Vect1 := [1, 2, 3, 4, 5]           Vect1 := [1, 2, 3, 4, 5]           (3.1.1)
```


```
> Vect2 := [3, 8, 6, -2, 8]        Vect2 := [3, 8, 6, -2, 8]           (3.1.2)
```

```
> plot(Vect1, Vect2, gridlines = true)
```



It is also possible to make 3D plots.

```
> plot3d(cos(x·y), x=-4..4, y=-4..4)
```



▼ Bode plots

Loading [DynamicSystems](#)


Using the DynamicSystems package, it is possible to convert transfer functions into "systems", whose bode plot can be very easily plotted.


```
> sys := TransferFunction(FrequencyDomain);
```

```
sys := [
  Transfer Function
  continuous
  1 output(s); 1 input(s)
  inputvariable = [uI(s)]
  outputvariable = [yI(s)]
]
```

(3.2.1)

```
> BodePlot(sys)
```





▼ Include/import data

▼ Import data from Excel file

▼ Embedded spreadsheet

▼ Import data from Maple file

